
Web Table 1: Risk of lung cancer mortality by categories of cumulative exposure and average exposure intensity.
	Exposure Metric
	N cases
	HRa (95% CI)
	aHRa (95% CI)

	Cumulative Exposure
	
	
	

	0 - <5 ppm-years
	92
	1 (ref)
	1 (ref)

	5 - <25 ppm-years
	48
	0.98 (0.68, 1.37)
	1.03 (0.71, 1.45)

	25 - <150 ppm-years
	32
	1.01 (0.67, 1.49)
	1.04 (0.68, 1.55)

	150+ ppm-years
	4
	0.54 (0.16, 1.29)
	0.53 (0.16, 1.29)

	
	
	
	

	Average Exposure Intensity
	
	
	

	0 - <15 ppm
	99
	1 (ref)
	1 (ref)

	15 - <45 ppm
	34
	0.99 (0.66, 1.45)
	1.12 (0.74, 1.64)

	45 - <70 ppm
	18
	1.07 (0.62, 1.72)
	1.2 (0.68, 2)

	70+ ppm
	25
	0.83 (0.52, 1.26)
	0.79 (0.49, 1.22)


HR = Hazard Ratio, aHR = adjusted Hazard Ratio
a –Both models match on attained age. The adjusted hazard ratio additionally matched on birth year (+/- 5yrs of index case), race, sex and controlled for SES prestige score with a spline. 


Web Appendix 1
The g-estimation procedure used in this study was guided by the appendices of Chevier et al (2012)1 and Hernan et al (2005)2. These papers provide a detailed discussion of the theory and motivation for the method. This appendix provides a step-by-step accounting of the g-estimation procedure that was applied to the boatbuilder cohort. 
The cohort included 5163 persons employed for at least one day at one of 2 Washington state boatbuilding facilities with no missing demographic or work history data. Appendix Table 1A and 1B give data formats and variable names for 3 hypothetical cohort members. The data had one row per person. Each worker may have held different jobs with different exposure values during their work history. These periods were encoded into a wide array into the person table. The example persons are shown with a maximum number of 2 exposure periods. 
Web Table 2: Example data format for 3 hypothetical cohort members
	ID
	Gender
	race
	VS
	dob
	pybegin
	dlo
	Facility
	
	Work History

	
	
	
	
	
	
	
	
	LC
	b1
	e1
	h1
	b2
	e2
	h2

	1
	M
	W
	Deceased
	7/1/1952
	9/18/1978
	5/31/1990
	A
	1
	9/18/1978
	10/1/1978
	32
	10/2/1978
	10/20/1978
	32

	2
	M
	W
	Deceased
	1/1/1935
	8/3/1966
	11/19/2012
	A
	0
	8/3/1966
	8/14/1966
	4
	8/15/1966
	9/9/1966
	48

	3
	F
	O
	Alive
	3/1/1932
	1/19/1972
	12/31/2016
	B
	.
	1/19/1972
	1/30/1972
	4
	 
	 
	 


Web Table 3: Data dictionary for the person file
	 
	ID
	person identifier

	
	gender
	M=male, F=female

	
	race
	W=White, O=other race

	
	VS
	vital status as of 12/31/2016

	
	dob
	date of birth 

	
	pybegin
	person year begin date

	
	dlo
	date last observed

	
	Facility
	Place of employment (Plant A or Pant B)

	
	LC
	indicator for lung cancer case (1=yes, 0=no)

	Work history 
information
	b1
	begin date of exposure period 1

	
	e1
	end date of exposure period 1

	
	h1
	daily exposure rate period 1 (ppm)

	
	b2
	begin date of exposure period 2

	
	e2
	end date of exposure period 2

	
	h2
	daily exposure rate period 2 (ppm)




First, helper variables were calculated using the SAS3 code below:
data person; set person;
 array b (2) b1-b2;
 array e (2) e1-e2;
 array h (2) h1-h2;

 age_beg_yr = (pybegin - dob)/365.25; *Age at person year begin date;
 age_end_yr = (dlo - dob + 1)/365.25; *Age at date last observed;

 yoh = year(pybegin); *Year of Hire;

 totdaysexp = 0;
 TA = idms_dlo - pybegin + 1; *Observed survival time;
 maxT = mdy(12,31,2016) - pybegin + 1; *Maximum possible survival time (i.e. time until study end date);
 do i=1 to 2;
   if h(i) > 30 then totdaysexp = totdaysexp + (e(i) - b(i) + 1); *Total days with exposure >30ppm;
 end;
run;
Next, there are two types of censoring that are addressed differently in this analysis: 1) administrative censoring, which includes those who are alive at the study end date (12/31/2016) and 2) non-administrative censoring, which includes those lost to follow-up and those who died from an outcome other than lung cancer. Individuals who are non-administratively censored are dropped from the analysis and inverse-probability weights (IPW) are calculated and applied to all subsequent analyses. Administratively censored individuals were addressed by adjusting the outcome variable of interest (as described below). In our analysis, there were 176 lung cancer cases, 3025 administratively censored and 1962 non-administratively censored individuals. 
IPW were calculated for non-administrative censoring by performing a logistic regression to calculate the probability of not being censored on each day of follow-up. Therefore, first a data-set was created with a row for each day of follow-up for each person. The following code creates a data-set with 73,224,884 observations, representing the 73,224,884 person-days (200,479 person-years) of follow-up for the 5163 cohort members (or, if applied to the hypothetical data in Table 1A would generate a data-set of 37,604 observations representing the 103 person-years of follow-up):
data person_day; set person; by id;
 keep id dlo date age race gender totdaysexp maxT TA exp daysexp censored emp dur rate a;
 array b (2) b1-b2;
 array e (2) e1-e2;
 array h (2) h1-h2;

 daysexp = 0;
 cumexp = 0;
 dur = 0;
 rate = 0;
 do i=1 to 2;
   if b(i) > . then do;
   do date=b(i) to e(i);
       exp = 0; if h(i) > 30 then exp = 1;
	 censored = 0; if date = dlo and LC not in (1) and dlo ne mdy(12,31,2016) then censored = 1;
	 age = (date - dob + 1)/365.25;
     	 emp = 1;
	 a = h(i);
	 output;
	 daysexp = daysexp + exp;
	 dur = dur + emp;
	 cumexp = cumexp + h(i);
	 rate = cumexp / dur;
   end;
   end;
 end;
 dle = date;
 do date = dle + 1 to dlo;
       exp = 0;
	 censored = 0;
	 age = (date - dob + 1)/365.25;
	 if date = dlo and LC not in (1) and dlo ne mdy(12,31,2016) then censored = 1;
	 emp = 0;
	 a = 0;
	 output;
	 daysexp = daysexp + exp;
 end;
run;
data person_day; set person_day;
 %rcspline(age, 25, 45, 65);
 dyr = date/365.25 + 1960;
 %rcspline(dyr, 1970, 1990, 2010);
 %rcspline(daysexp, 1, 400, 2000); 
run;

The number of days with exposure >30ppm (daysexp), currently employed (emp), duration employed (dur), average exposure rate (rate) and current exposure level (a) were time dependent and evaluated on each day of follow-up. In addition, censored was a 0/1 indicator indicating the last day of follow-up for persons either lost to follow-up or who died of an outcome other than lung cancer. Continuous variables were entered into the logistic regression as a restricted cubic spline using Frank Harrell’s macro (found here: http://biostat.mc.vanderbilt.edu/wiki/pub/Main/SasMacros/survrisk.txt). 
Logistic regression on this data to estimate the probability of not being censored given demographics and exposure history. The predicted probabilities (p0) on each day are then outputted:
*IPW (per SAS log, this models P(censored = 0)...i.e. probability of not being censored);
proc logistic data= person_day;
 class race gender;
 model censored = age age1 dyr dyr1 race gender daysexp daysexp1 rate rate1 emp p_score p_score1;
 output out= person_day predicted=p0;
run;

The results of this regression are below (the parameter estimates as well as graphs of splines for the continuous variables):



Web Table 4: Parameter estimates of logistic regression modeling probability of not being censored on any given day.Web Figure 1: Plots of the spline parameter estimates for the continuous measures in the logistic regression modelling the probability of not being censored on any given day

Once probabilities are calculated, the IPW is calculated by taking the cumulative product of each probability of each individual over time and then taking the inverse. In addition, the 2476 non-administratively censored are dropped. Furthermore, all future analyses are restricted to employed (emp=1) person time. 

**Calculate IPW;
proc sort data= person_day; by id descending date; run;
data person_day; set person_day; by id;
 retain cump;
 if first.id then do;
   cump = 1;
 end;
 IPW = 1/cump;
 output;
 cump = cump*p0;

 *Cap IPW to 10;
 if IPW > 10 then IPW = 10;
run;

**Drop non-admin censored members and restrict to employed person-time;
data censored; set person_day;
 keep id;
 if censored = 1;
run;
data employed_pt; merge person_day (in=_a) censored (in=_b); by id;
 if _a and not _b;
 if emp = 1;
run;

This final data-set contains the 1,513,480 observations from the employed person-time of the 176 lung cancer cases, 3025 administratively censored cohort members. 
The final step in the analysis is to perform a grid search of phi values and to then calculate counter-factual survival times. The correct value of phi is the one that calculates counter-factual survival times that are independent of exposure, given demographics and past exposure history. The below walks through this analysis for one value of phi (phi =1). 
To address administrative censoring, a new variable is calculated (coded as D here and referred to as D in the appendices of Chevier et al (2012)1 and Hernan et al (2005)2). D is a 1/0 indicator variable indicating whether all possible potential survival times under no exposure will be observed. This can be checked by checking the extreme potential survival times of always being exposed (Talways) and never being exposed (T0). D will be 1 if both Talways and T0 extend to before the study end date of 12/31/2016 (i.e. T0 and Talways are less than maxT_running). 
%let phi=1;
data employed_pt; set employed_pt; by id;
 retain totdaysexp_running TA_running maxT_running;
 if first.id then do;
   totdaysexp_running = totdaysexp;
 end;
 TA_running = dlo - date + 1;
 maxT_running = mdy(12,31,2016) - date + 1; 

 T0 = totdaysexp_running*exp(&phi) + (TA_running - totdaysexp_running); 
 Talways = T0*exp(-&phi);
 if T0 < maxT_running and Talways < maxT_running then D = 1; else D = 0;
 %rcspline(maxT_running, 16000, 17000, 19000);

 totdaysexp_running = totdaysexp_running - exp;
run;

A logistic regression is then run to calculate the relationship of D on the probability of being exposed (an exposure level >30ppm) on any given day. The coefficient of D and its significance are of interest. The correct value of phi is one in which D is independent of exp (i.e. its coefficient is 0 and not significant). 
proc logistic data=employed_pt  outest=beta covout;
 weight IPW;

 class race gender;
 model exp = age age1 dyr dyr1 race gender daysexp daysexp1 rate rate1 p_score p_score1 
             D maxT_running maxT_running1;
run;
A macro was written repeating the above two steps for different values of phi (from -2 to 2). Below plots the Wald-Z statistic of D for various values of phi:





Web Figure 2: Plot of the Wald-Z statistic for the D parameter obtained from the grid-search.
[image: ]
The dotted lines are at -1.96 and +1.96 and are used to calculate the confidence intervals of phi. The variable Dsum represents the number of persons with an observed counter-factual survival time (i.e. D=1). This helps to put into perspective the points of stability of this graph. 
We note that this method of assigning confidence intervals relies on asymptotic normality assumptions which may not be valid for small samples sizes. Bootstrapping may provide more accurate coverage; however, bootstrapping comes with practical challenges. Namely, the approach in this paper was to visually inspect the above graph for points where the plot crosses 0. This would have to be automated. Additionally, in a given sample there may be multiple point estimates found, or none in the chosen range of the grid-search. Furthermore, there is also no guarantee a solution will be found in any given sample.
Given the nature of this cohort, the grid search procedure is highly variable for values of psi less than 0. This is because most cohort members had a very short duration of exposure (median was 5 months). Therefore, during the g-estimation procedure, the counterfactual exposure scenario of ‘always exposed’ dramatically changes the associated counterfactual survival times from what was observed. When psi is negative (i.e. exposure is protective) the counterfactual survival times get dramatically extended and become administratively censored (i.e. survival extends past the study end date of 12/31/2016). This is represented by the small Dsum values in the plots, and the minimum Dsum in these plots is as small as 8 observed lung cancer cases. 

As psi increases, the number of observed cases increases. For negative values of psi, each addition has a dramatic effect on results. In fact, it may be misleading to connect these points with a line, because the function might not be continuous everywhere. There is some “jumping around” from positive to negative.   

With this in mind, as well as from additional support from the below sensitivity analyses, we feel the positive estimate is potentially the “true” estimate. 



Web Appendix 2
Various sensitivity analyses were performed to test the robustness of the above result. 
Current Employment
First, persons currently employed during record collection in October of 1978 were assumed to be unexposed for the remainder of their follow-up through December 31, 2016 (as opposed to having their person-time ending on October 1, 1978). Below is the plot of this analysis:
Web Figure 3: Plot of the Wald-Z statistic for the D parameter obtained from the grid-search for the model with current employed workers’ follow-up extended.
[image: ]
[bookmark: _Hlk57638107]This yielded an estimate for phi of 1.30 (95% CI: 1.23, 1.35).

Employed greater than 3 months
An additional analysis considered applying a minimum duration of employment to the cohort criterion. We restricted to person-time with more than 3 months of employment yielding the below results:
Web Figure 4: Plot of the Wald-Z statistic for the D parameter obtained from the grid-search restricting to person-time with >3 months of employment.
[image: ]
Which corresponds to an estimate of phi of 1.50 (95% CI: 1.44, 1.54).

Removal of Sampling Weights
The effect of removing the censoring weights calculated from the inverse probability weighting on the exposure was modeled. We note that the weights are used to reduce selection bias from dropping non-administratively censored individuals as discussed in Hernan et al (2005) and Chevier et al (2012), and further clarified in Tchetgen Tchetgen et al (2012), , however, this analysis simply highlights the sensitivity of the results on how the censoring model was specified. Below are the results:
Web Figure 5: Plot of the Wald-Z statistic for the D parameter obtained from the grid-search with sampling weights removed from the logistic regression.
[image: ]
This yielded an estimate for phi of 1.73 (95% CI: 1.68, 1.82).

Model Flexibility 
Finally, the effect of the flexibility of the exposure and censoring models was evaluated. In particular, a restricted cubic spline with 3 knots was used for all continuous variables in the original model described above. We tested the effect of increasing the number of knots from 3 to 5 equally spaced knots. Below is the plot of this analysis:
Web Figure 6: Plot of the Wald-Z statistic for the D parameter obtained from the grid-search.
[image: ]
This yields an estimate of phi of 1.72 (95% CI: 1.66, 1.78). The g-estimation procedure requires both the censoring model and exposure model to be correctly specified, and in that regard, a more flexible model is preferable. In addition, this model yields a plot in which the graph only crosses 0 once. However, we chose to present the original results in the main text since it was our a priori model (this more flexible model wasn’t investigated until journal reviewer comments) and the previous results are more conservative. 

